

Álgebra Lineal 2 - Taller 5

Universidad de los Andes - Departamento de Matemáticas

Para cada uno de los siguientes operadores $f \in \text{Hom}_K(V, V)$, $V = K^d$:

i) $f \in \text{Hom}_{\mathbb{Q}}(\mathbb{Q}^3, \mathbb{Q}^3)$ donde

$$f(x, y, z) = (4x, 3x + y - 3z, 3x - 3y + z),$$

ii) $f \in \text{Hom}_{\mathbb{Q}}(\mathbb{Q}^3, \mathbb{Q}^3)$ donde

$$f(x, y, z) = \frac{1}{2}(5x - y + z, 5x - y - 3z, 4x - 4y)$$

en la base canónica \mathcal{C} ,

iii) $f \in \text{Hom}_{\mathbb{Q}}(\mathbb{Q}^4, \mathbb{Q}^4)$ donde

$$f(x, y, z, w) = \frac{1}{3}(11x - y - z + 2w, 14x + 2y - 7z - 4w, 13x - 5y + z - 5w, 13x - 8y - 5z + 4w)$$

iv) $f \in \text{Hom}_{\mathbb{Q}}(\mathbb{Q}^4, \mathbb{Q}^4)$ donde

$$f(x, y, z, w) = \frac{1}{3}(-2x + y + 4z - 2w, 3x - 6y + 6z, 2x - 4y + 5z + 2w, -6y + 6z + 3w),$$

v) $f \in \text{Hom}_{\mathbb{R}}(\mathbb{R}^4, \mathbb{R}^4)$ donde

$$f(x, y, z, w) = \frac{1}{3}(-2x + y + 4z - 2w, 3x - 6y + 6z, 2x - 4y + 5z + 2w, -6y + 6z + 3w),$$

vi) $f \in \text{Hom}_{\mathbb{Q}}(\mathbb{Q}^4, \mathbb{Q}^4)$ donde

$$f(x, y, z, w) = \frac{1}{3}(9x - 3y - 6w, 3x - 3y + 6z, 8x - 10y + 5z - w, 10x - 8y + z + w),$$

vii) $f \in \text{Hom}_{\mathbb{C}}(\mathbb{C}^4, \mathbb{C}^4)$ donde

$$f(x, y, z, w) = \frac{1}{3}(9x - 3y - 6w, 3x - 3y + 6z, 8x - 10y + 5z - w, 10x - 8y + z + w),$$

1. Encuentre la representación matricial de f en la base canónica.
2. Encuentre la factorización del polinomio característico $P_f(t)$ como producto de potencias de polinomios irreducibles de $K[t]$:

$$P_f(t) = \prod_{i=1}^n P_i(t)^{r_i}$$

donde cada $P_i(t) \in \mathbb{K}[t]$ es irreducible.

3. Sean $V_i = \ker(P_i(f)^{r_i})$, $i = 1, \dots, n$, y p_1, \dots, p_n las proyecciones sobre V_1, \dots, V_n respecto a la descomposición $V = \bigoplus_{i=1}^n V_i$. Encuentre polinomios $\Pi_1(t), \dots, \Pi_n(t) \in K[t]$ tales que $\Pi_i(f) = p_i$, $i = 1, \dots, n$.
4. Encuentre una base de V respecto a la cual la representación matricial de f es diagonal por bloques de tamaño $\dim(V_i)$, $i = 1, \dots, n$, y cada bloque corresponde a una representación matricial de $f|_{V_i}$.

Ejemplo

Sea

$$A = \begin{bmatrix} 2 & 1 & -1 \\ 0 & 2 & 0 \\ 0 & 1 & 1 \end{bmatrix}$$

Para calcular su polinomio característico en la variable t y su factorización ejecutamos:

```
sage: t=PolynomialRing(RationalField(), 't').gen()
sage: A=Matrix(QQ, [[2,1,-1],[0,2,0],[0,1,1]])
sage: P=A.charpoly('t')
sage: factor(P)
```

Vemos que $P_A(t) = (t-1)(t-2)^2$. Tenemos dos subespacios invariantes $V_1 = \ker(A - I_3)$ y $V_2 = \ker((A - 2I_3)^2)$. Para obtener los polinomios $\Pi_1(t), \Pi_2(t) \in \mathbb{Q}[t]$ tales que, para $i = 1, 2$, $\Pi_i(A)$ sea la proyección sobre V_i , ejecutamos

```
sage: P1=(t - 1)
sage: P2=(t - 2)^2
sage: P12,Q1,Q2=xgcd(P1,P2)
sage: Pi1=Q2*P2
sage: Pi2=Q1*P1
```

Las dos matrices de proyección $\Pi_i(A)$, $i = 1, 2$, las calculamos ejecutando:

```
sage: Pr1=Pi1(A)
sage: Pr2=Pi2(A)
```

Para verificar que obtenemos dos matrices de proyecciones con las características deseadas ejecutamos:

```
sage: Pr1^2-Pr1
sage: Pr2^2-Pr2
sage: Pr1+Pr2
```

Sea

$$A = \begin{bmatrix} -1 & -2 \\ -1 & 1 \end{bmatrix}$$

Para calcular su polinomio característico en la variable t y su factorización ejecutamos:

```
sage: t=PolynomialRing(QQ, 't').gen()
sage: A=Matrix(QQ, [[-1,-2],[-1,1]])
sage: P=A.charpoly('t')
sage: factor(P)
```

Vemos que $P_A(t) = t^2 - 3$. Para factorizarlo en los reales (donde $P_A(t) = (t - \sqrt{3})(t + \sqrt{3})$) debemos agregarle en Sage $\sqrt{3}$ a nuestro cuerpo \mathbb{Q} . Para ello ejecutamos:

```
sage: K.<r> = QQ.extension(P)
```

Esto le agrega la raíz de $P_A(t)$ a nuestro cuerpo. El nuevo cuerpo se llama K y en él r denota $\sqrt{3}$. Para factorizarlo debemos explicarle a Sage que ahora vamos a trabajar en K en vez de \mathbb{Q} .

```
sage: P=P.change_ring(K)
sage: factor(P)
```

Tenemos dos subespacios invariantes $V_1 = \ker(A - \sqrt{3}I_2)$ y $V_2 = \ker(A + \sqrt{3}I_2)$. Para obtener los polinomios $\Pi_1(t), \Pi_2(t) \in \mathbb{R}[t]$ tales que, para $i = 1, 2$, $\Pi_i(A)$ sea la proyección p_i sobre V_i de acuerdo a la descomposición $\mathbb{R}^3 = V_1 \oplus V_2$, ejecutamos

```
sage: P1=(t - r)
sage: P2=(t + r)
sage: P12,Q1,Q2=xgcd(P1,P2)
sage: Pi1=Q2*P2
sage: Pi2=Q1*P1
```

Las dos matrices de proyección $\Pi_i(A)$, $i = 1, 2$, las calculamos ejecutando:

```
sage: Pr1=Pi1(A)
sage: Pr2=Pi2(A)
```

El espacio $V_1 = \text{im}(p_1)$ está generado por $v_1 = (-\sqrt{3} + 3, -\sqrt{3})$ y $V_2 = \text{im}(p_2)$ está generado por $v_2 = (\sqrt{3} + 3, \sqrt{3})$. Hacemos un cambio de base para ver la representación matricial de $f(x, y) = (-x - 2y, -x + y)$ en la base $\{v_1, v_2\}$:

```
sage: v1=vector(K, [-r+3, -r])
sage: v2=vector(K, [r+3, r])
sage: C=column_matrix([v1,v2])
sage: B=C^(-1)*A*C
```

Note que si trabajamos directamente sobre los reales en Sage, en vez de pedirle que use el símbolo r para $\sqrt{3}$, tomará valores aproximados para $\sqrt{3}$. Esto lo puede verificar ejecutando

```
sage: t=PolynomialRing(RR, 't').gen()
sage: A=Matrix(RR, [[-1, -2], [-1, 1]])
sage: P=A.charpoly('t')
sage: factor(P)
```

Para la pregunta *vii*) lo mejor es trabajar con $K = \mathbb{Q}[i]$ esto se logra definiendo K por

```
sage: K.<i> = QQ.extension(t^2+1)
```

Si usted trabaja directamente sobre los complejos, como con los reales, Sage tomará aproximaciones en la parte real y la parte compleja.